COBL Combo Disk Boot Loader User's Guide

Version 3.00 Martin Eberhard Mike Douglas 16 January 2016

Revision History

Revision	Date	Author	Notes
1.00	1 May 2014	M. Eberhard	Created from DBLe and MDBL, to produce exactly either of these loaders, selected by an assembly option.
2.00	7 May 2014	M. Eberhard	General code rewrite. Automatic detection of disk drive type. Select disk unit via sense switches. Improve restore. Add 'O' error. Verify after code copy.
2.01	15 May 2014	M. Eberhard	Step in once before seeking track 0, in case the head is out past track 0. Restart 6.4 Sec motor shutoff timer on retries.
2.02	4 June 2014	M. Eberhard	Eliminate booting from other than drive 0, because most Altair software just loads a 2-sector loader, and that loader loads the rest of the code (always from drive 0).
2.03	17 Jan 2015	M. Douglas	Add 40 mS delay prior to changing seek direction, to meet Pertec FD400 spec
2.04	12 Mar 2015	M. Douglas	Fix 88-2SIO initialization value
2.05	11 Jan 2016	M. Eberhard	No IN or OUT instructions until the code is moved into RAM, for compatibility with some Turnkey modules. Also fix bug when reporting an overlay error.
3.00	16 Jan 2016	M. Douglas	Make PROM position independent

CONTENTS

1.	INTRODUCTION	3
2.	INSTALLATION	4
3.	MEMORY REQUIREMENTS	4
4.	SENSE SWITCH SETTINGS	4
5.	OPERATING PROCEDURES	5
6.	ERROR INDICATIONS	5
API	PENDEX A - ALTAIR DISK BOOT-TRACK FORMAT	7
API	PENDIX B - SOURCE CODE LISTING	8

1. INTRODUCTION

CDBL is a boot loader PROM program for booting from either the Altair 88-DCDD 8" floppy disk system or the Altair 88-MDS Minidisk system.

CDBL works exactly the same as the Altair DBL boot PROM (for the 88-DCDD 8" disk controller) and the Altair MDBL boot PROM (for the 88-MDS Minidisk controller), with the following improvements:

1. Automatic Disk Drive Type Detection

Before booting, CDBL determines whether the disk is an 8" disk or a Minidisk by identifying the sector that follows sector 15. Sector 16 means it is an 8" disk (which has 32 sectors per track), while sector 0 means it a Minidisk (which has only 16 sectors per track).

2. RAM Usage

Like DBL and MDBL, CDBL requires some zero wait-state RAM for fast code, a buffer, and the stack. DBL uses about 400 bytes starting at address 026000 octal (which limits the loaded program to no more than 11K bytes). MDBL uses about 400 bytes starting at 046000 octal (which limits the loaded program to no more than 19K bytes). CDBL uses 512 bytes starting at address 046000 octal, which limits the loaded program to 19K bytes.

3. Position Independence

CDBL version 3.00 and later will run from any PROM socket at a 256byte page boundary, except page 0.

4. Memory Overlay Detection

CDBL checks to see if the disk data will overwrite CDBL's own RAM (at 046000 octal), and will abort with an overlay error if so.

5. Track 0 Overshoot Correction

CDBL steps inward once, before seeking track 0. This will allow booting even when the track 0 end-stop is maladjusted. (On a maladjusted drive, it is possible to seek outward past track 0.)

6. Restart Shutoff Timer on Retries

The Altair Minidisk controller has a timer that gets restarted whenever a Step command is issued, or when an explicit Timer Reset command is issued. 6.4 seconds later, this timer shuts off the disk controller, unless it gets restarted first.

MDBL never explicitly restarts this timer, relying on the Step commands to restart the timer at the beginning of each track. However, each sector-read retry will add 200 mS to the load time of the current track. This means that 32 retries on one track will cause the MDBL load to fail because the controller will shut off. Unlike MDBL, CDBL restarts this timer whenever it retries a sector read, so that it can complete booting even from a disk that requires many retries.

7. Correct 88-2SIO Initialization

The 88-2SIO initialization bug in the Altair DBL PROM has been corrected. This means that error codes will be printed correctly on an 88-2SIO Terminal. (This bug does not exist in the MDBL PROM.)

8. 88-2SIO Initialization Reset Delay

CDBL delays before resetting the 88-2SIO long enough that the transmission of any character in its output buffer (e.g. the 'B' command from UBMON) will complete.

2. INSTALLATION

Like the DBL and MDBL PROMs, the CDBL PROM is normally installed at address 177400 octal. It belongs in slot H on an 88-PMC memory card (which is jumpered to start at address 174000 octal), or in slot H1 on a Turnkey Module.

However, CDBL (version 3.00 or later) will run from any PROM socket. Note that UBMON expects it to be at address 174000 octal. Examples in this manual assume 174000 octal - substitute the start address of your PROM socket if your CDBL is installed elsewhere.

3. MEMORY REQUIREMENTS

CDBL requires two 256-byte pages of zero wait-state RAM starting at address 046000 octal. The code that is loaded from the disk gets written to RAM, starting at address 0. For practical purposes, this means that the Altair must have at least 20K bytes of RAM starting at address 0, and at least the last 4K block of this RAM must be zero wait-state RAM. (This zero wait-state RAM may be dynamic RAM, as the timing has sufficient margin to handle the occasional refresh cycle.)

Because of the location of CDBL's RAM page, the boot file that is loaded from the disk cannot be larger than 19K bytes in size. (This is the same boot file size limitation as MDBL. DBL limits boot files to 11K bytes.)

4. SENSE SWITCH SETTINGS

Loaded software, such as Altair Disk BASIC, will normally read the four sense switches <A15:A12> to determine the Terminal device. See the manual for the software that you are loading.

5. OPERATING PROCEDURES

5.1 ACCESSING CDBL

If you do not have TURMON or UBMON, then use the front panel to STOP and RESET the Altair, and then EXAMINE address 177400 octal.

5.2 SETTING THE SENSE SWITCHES

 Set Sense Switches <A15:A12> according to the manual for the program you are loading (e.g. the Altair BASIC Reference manual, Appendix B). For reference, here are the standard Altair Terminal Device Sense Switch settings for many Altair programs:

Terminal Device	A15	A14	A13	A12
88-2SIO Port 0	0	0	0	0
(2 stop bits)				
88-2SIO Port 0	0	0	0	1
(2 stop bits)				
88-SIO	0	0	1	0
4PIO	0	1	0	0
PIO	0	1	0	0

5.3 INITIATING THE BOOT

- Insert the boot disk into the disk drive. If you are using a Minidisk, check that the disk has a write-protect tab installed.
- 2. Start CDBL
 - a. If you are using TURMON, type "J177400" on the Terminal.
 - b. If you are using UBMON, type "B" on the Terminal.
 - c. If you are starting with a front panel, press the RUN switch.

5.4 BOOTING

CDBL will read the Altair disk boot file one sector at a time, starting with track 0, sector 0, and write the file into RAM, starting at address 0. Within each track, the sectors are interleaved 2:1 - the even sectors are loaded first, followed by the odd sectors.

Loading will continue until either an irrecoverable error is encountered, or until the complete boot file has been loaded into RAM. If the file load completes, then CDBL will jump to address 0 to execute the loaded code. If an irrecoverable error occurs, then CDBL will report that error and hang - see the next section.

Note that many programs (including later versions of Altair Disk Basic) load just a few sectors from disk, and then jump to that little loader program to load the rest of the software.

6. ERROR INDICATIONS

The front panel's Interrupt Enable light remains off while loading is proceeding properly. If CDBL encounters an irrecoverable error, it will turn on the Interrupt Enable light, store the ASCII error code in memory location 0, and store the 16-bit offending address in memory locations 1 and 2. CDBL will then send the error code to all standard Altair Terminal devices continuously until you STOP or RESET the Altair from the front panel.

The CDBL error codes are:

- C Checksum error-----A computed checksum and the checksum on the disk did not match, or a Marker Byte was not 377 octal. CDBL will retry any sector that has either of these errors 15 times before giving up and indicating a Checksum error.
- M Memory error-----Defective memory, read-only memory, or protected memory was encountered when attempting to write to RAM; the address of the offending RAM location is stored in memory locations 1 and 2.
- O Overlay error-----An attempt was made to load disk data beyond the first 19K bytes of memory, which would overlay the memory page that contains CDBL's stack, buffer, and disk routines.

APPENDEX A - ALTAIR DISK BOOT-TRACK FORMAT

Altair disks are hard-sectored disks. 8" disks (for the 88-DCDD) have 32 sectors per track, while Minidisks (for the 88-MDS system) have 16 sectors per track. Sectors are numbered sequentially on each track, starting at sector 0. Sectors each contain 128 bytes of actual data.

Tracks are numbered sequentially from track 0, starting at the outside disk diameter. 8" disks have 77 tracks, and Minidisks have 35 tracks. Loading begins with track 0, and continues sequentially through the tracks until the required amount of data (which is specified in the File Byte Count portion of each sector's header) has been read. The first several tracks are reserved for the boot code, and usually have a slightly different format than the rest of the disk.

For at least the boot portion of an Altair disk, sectors are interleaved with a 2:1 pattern: on each track, the even sectors (starting with sector 0) are all loaded first, followed by the odd sectors (starting with sector 1). (CDBL requires less than one sector time to process a sector's data once it has been read into the buffer. So with 2:1 interleave, each track can be read with just two disk revolutions, once the track step is complete and sector 0 has been found.)

Each sector contains 137 bytes. On the boot tracks, these sectors are formatted as follows:

Byte O	Track Number, with MSB set (the sync bit)
Bytes 1-2	File Byte Count (16 bits)
Bytes 3-130	Sector Data (128 bytes)
Byte 131	Marker Byte, must be 377 octal
Byte 132	Sector Data Checksum (8-bit sum of all data bytes)
Bytes 133-136	Spare bytes (not read by CDBL)

January 16, 2016 88 CDBL PROM vers. 3.00

APPENDIX B - SOURCE CODE LISTING

The following pages list the source code for CDBL. This code was assembled using Digital Research's ASM assembler. As such, all values are in hexadecimal, rather than in octal as is normal for MITS software.

CDBL.PRN

:======================================	
;= For the Altair 88-D0	sk Boot Loader ROM = CDD 8" disk system and = DS Minidisk system = Eberhard =
	Altair Disk BASIC) from an = an 88-MDS 5-1/4" minidisk, = ich kind of drive is attached. =
;=====================================	ΓES =
;= ;= Like DBL and MDBL, CDBL fi ;= EPROM and execution begins ;= since version 3.00, the PRG ;= and can run at most any 250	at address FF00h. However, = DM is position independent = 5 byte boundary. =
;= some versions of MITS's 880 ;= when any IN instruction is ;= into RAM at 4C00h (RAMADR)	<pre>= = = = = = = = = = = = = = = = = = =</pre>
;= Minidisks have 16 sectors/t ;= 8" disks have 32 sectors/t ;= CDBL figures out which kind ;= based on the existance of s	<pre>track, numbered 0 through 15. = rack, numbered 0 through 31. = d of disk drive is attached, =</pre>
;= ALTAIR DISK SECTOR FO	DRMAT (FOR BOOT SECTORS) =
;= BYTE(s) FUNCTION ;= 0 Track number+80 ;= 1 File size low B ;= 2 File size High ;= 3-130 Sector data ;= 131 Marker Byte (OF ;= 132 Checksum ;= 133-136 Spare	byte RAMADR+7Ch = Byte RAMADR+7Dh = RAMADR+7Eh to RAMADR+FDh =
;= Each sector header contains ;= this many bytes (rounded up ;= from the disk and written ;= When done (assuming no error ;= address 0 (DMAADR) to execu	s a 16-bit file-size value: = b to an exact sector) are read = to RAM, starting at address 0. = brs), CDBL then jumps to = ute the loaded code. =
;= on each track first (start ;= followed by the odd sectors ;= continuing through the into ;= until the specified number	s (starting with sector 1), = erleaved sectors of each track =
;= bytes) from disk into the F ;= checks to see if this sector ;= portion of CDBL, and aborts ;= then copies the data paylog ;= its final RAM location, ca ;= way. During the copy, each	or (including the actual data = neader and the first 2 trailer = RAM buffer (SECBUF). Next, CDBL = or would overwrite the RAM = s with an 'O' error if so. It = ad portion from the buffer to = lculating the checksum along the =

CDBL.PRN ;= abort the load with an 'M' error. = ;= = ;= Any disk read error (a checksum error or an incorrect ;= marker byte) will cause a retry of that sector read. After = ;= 16 retries on the same sector, CDBL will abort the load ;= with a 'C' error. = _ = = := If the load aborts with any error, then CDBL will turn on = ;= the INTE LED on the front panel (as an indicator), write = ;= the error code ('C', 'M', or 'O') to RAM address 0, write = ;= the offending memory address to RAM addresses 1 and 2, and = = then hang forever in a loop, printing the error code to = = all known Altair Terminal output ports. = :== **REVISION HISTORY** = = ' = = 1.00 05May2014 M.Eberhard = = Combined MDBL and DBLme code, with assembly options to := create exactly both of these boot loaders := = 2.00 08May2014 M.Eberhard = = Restructure and squeeze the code. Automatic 8"/Minidisk = detection. Select boot disk from the sense switches. = := = Improve track 0 seek by waiting for -MVHEAD before testing TRACK0. Detect memory overrun errors. Verify = = = = copy of CDBL code into RAM. = = 2.01 15May2014 M.Eberhard := = Step in once before seeking track 0. Restart 6.4 Sec := = ;= motor shutoff timer on retries. = 2.02 04Jun2014 M. Eberhard ;= = ;= Eliminate booting from other than drive 0 because Basic = and Burcon CP/M just load a 2-sector boot loader, and that boot loader loads the rest, always from drive 0. 2.03 17Jan2015 M. Douglas := := = = = Force 43ms minimum delay when changing seek direction = = to meet/exceed 8" drive requirements. = 2.04 12Mar2015 M. Douglas := ;= Change 2SIO init constant (ACINIT) from 21h (7E2, xmit = interrupts on) to 11h (8N2) 05 11Jan2016 M. Eberhard No IN or OUT instructions until code is moved to RAM = = 2.05 = = := = (for compatibility with some Turnkey Modules). This = increased the RAM footprint from 256 to 512 bytes. Also = = = fixed a bug when reporting overlay errors. = = 3.00 12Jan2016 M. Douglas := = Make the PROM position independent by making the := = routine that copies PROM to RAM position independent. = = _____ _____ Disk Parameters _ _ _ _ _ _____ BPS 128 ;data bytes/sector equ MDSPT 16 :Minidisk sectors/track equ ;this code assumes SPT for 8" ;disks = MDSPT * 2. HDRSIZ equ 3 ;header bytes before data 2 trailer bytes read after data TLRSIZ equ SECSIZ equ BPS+HDRSIZ+TLRSIZ ;total bytes/sector RETRYS equ ;max retries per sector 16

0080 =

0010 =

0003 =

0002 =

0085 =

0010 =

CDBL.PRN

	•	CDBL.PRN		
	; about these v ; 1) RAMADR ar ; 2) The addre ; sector bu	values are embedd nd PROM low addre ess of the last b uffer) must be XX	ess byte = 0 byte of SECBUF (the SECSIZ-sized	
4C00 = 4D7B = 4D7B = 0000 =	SECBUF equ	RAMADR+512-SECS	;Address for code copied to RAM SIZ ;grows down from here ;Disk load/execution address	
	,	sector component		
4D7C = 4D7E = 4DFE = 4DFF =	SDATA equ	SECBUE+HDRST7	;address of file size ;address of sector data ;address of marker byte ;address of checksum byte	
	, 88-SIO Equate	 25 		
0000 = 0000 = 0001 =	, SIOCTL EQU SIOSTA EQU SIODAT EQU	00 00 01	;Control ;Status ;Rx/Tx Data	
	, 88-2SIO's por Equates (all	rt O, Turnkey Mod based on the Mot	dule, and 88-UIO corola 6850 ACIA)	
0010 = 0010 = 0011 =	ACCTRL equ ACSTAT equ ACDATA equ	10h 10h 11h	;ACIA Control output port ;ACIA Status input port ;ACIA Tx/Rx Data register	
0003 = 0011 =	ACRST equ ACINIT equ	03h 11h	;Master reset ;/16, 8bit, No Parity, 2Stops	
	; ; 88-PIO Equate	 25		
0004 = 0004 = 0005 =	; PIOCTL EQU PIOSTA EQU PIODAT EQU	04	;Control ;Status ;Tx/Rx Data	
	; ; 88-4PIO equat	 Ces		
0020 = 0021 = 0022 = 0023 =	P4CAO equ P4DAO equ P4CBO equ P4CBO equ P4DBO equ	20h 21h 22h 23h	;Port 0 Section A Ctrl/Status ;Port 0 Section A Data ;Port 0 Section B Ctrl/Status ;Port 0 Section B Data	
002C =	P4CINI equ	2Ch	;Control/status initialization	
	, Altair 8800 Disk Controller Equates (These are the same for the 88-DCDD controller and the 88-MDS controller.)			
0008 = 0080 =	, DENABL equ		;Drive Enable output ;disable disk controller	

CDBL.PRN

0008 = 0001 = 0002 = 0004 = 0008 = 0020 = 0040 = 0080 =	DSTAT ENWDAT MVHEAD HDSTAT DRVRDY INTSTA TRACKO NRDA	equ equ equ equ equ equ equ equ	08H 01h 02h 04h 08h 20h 40h 80h	;Status input (active low) ;-Enter Write Data ;-Move Head OK ;-Head Status ;-Drive Ready ;-Interrupts Enabled ;-Track O detected ;-New Read Data Available
0009 = 0001 = 0002 = 0004 =	DCTRL STEPIN STPOUT HDLOAD	equ	09h 01H 02H 04H	;Drive Control output ;Step-In ;Step-Out ;8" disk: load head ;Minidisk: restart 6.4 S timer
0008 = 0010 = 0020 = 0080 =	HDUNLD IENABL IDSABL WENABL	equ equ equ equ	08h 10h 20h 80h	;unload head (8" only) ;Enable sector interrupt ;Disable interrupts ;Enable drive write circuits
0009 = 0001 = 003E =	DSECTR SVALID SECMSK		09h 01h 3Eh	;Sector Position input ;Sector Valid (1st 30 uS ;of sector pulse) ;Sector mask for MDSEC
000A =	DDATA	-	0Ah	;Disk Data (input/output)
	;	e-byte e	rror messages	
0043 = 004D = 004F =	CERMSG MERMSG OERMSG	equ equ equ	'C' 'M' 'O'	;checksum/marker byte error ;memory write verify error ;Memory overlay error message
4C00	,	ORG	RAMADR	;assemble at dest RAM address
4-00 -2	,			
4C00 F3		ui		;turn off INTE (no error yet)
4C00 F3	; is	the PROM	independent so	for execution. This copy routine the boot PROM can be at most any M address and RAMADR must be 0.
4C00 F3 4C01 110E4C	; is	the PROM	independent so	for execution. This copy routine the boot PROM can be at most any
	; is	the PROM position ress. Th	independent so e LSB of the PRO	for execution. This copy routine the boot PROM can be at most any M address and RAMADR must be 0.
4C01 110E4C 4C04 317B4D 4C07 21E1E9 4C0A E5	; is	the PROM position ress. Th lxi lxi lxi push	independent so e LSB of the PRO d,MLOOP sp,STACK h,0E9E1h h	for execution. This copy routine the boot PROM can be at most any M address and RAMADR must be 0. ;DE->MLOOP in RAM ;H=PCHL,L=POP H ;POP H, PCHL at STACK-2, STACK-1
4C01 110E4C 4C04 317B4D 4C07 21E1E9 4C0A E5 4C0B CD794D 4C0E 3B	; is ; add ;	the PROM position ress. Th lxi lxi lxi push call dcx	independent so e LSB of the PRO d,MLOOP sp,STACK h,0E9E1h h STACK-2 sp	<pre>for execution. This copy routine the boot PROM can be at most any M address and RAMADR must be 0. ;DE->MLOOP in RAM ;H=PCHL,L=POP H ;POP H, PCHL at STACK-2, STACK-1 ;addr of MLOOP in HL and stack RAM ;point SP to MLOOP address</pre>
4C01 110E4C 4C04 317B4D 4C07 21E1E9 4C0A E5 4C0B CD794D 4C0E 3B 4C0F 3B 4C0F 3B	; is ; add ;	the PROM position ress. Th lxi lxi lxi push call dcx dcx dcx mov	independent so e LSB of the PRO d,MLOOP sp,STACK h,OE9E1h h STACK-2 sp sp a,m	<pre>for execution. This copy routine the boot PROM can be at most any M address and RAMADR must be 0. ;DE->MLOOP in RAM ;H=PCHL,L=POP H ;POP H, PCHL at STACK-2, STACK-1 ;addr of MLOOP in HL and stack RAM ;point SP to MLOOP address ; in stack memory ;get next EPROM byte</pre>
4C01 110E4C 4C04 317B4D 4C07 21E1E9 4C0A E5 4C0B CD794D 4C0E 3B 4C0F 3B 4C0F 3B 4C10 7E 4C11 12 4C12 1C	; is ; add ;	the PROM position ress. Th lxi lxi lxi lxi call dcx dcx dcx mov stax inr	independent so e LSB of the PRO d,MLOOP sp,STACK h,0E9E1h h STACK-2 sp sp a,m d e	<pre>for execution. This copy routine the boot PROM can be at most any M address and RAMADR must be 0. ;DE->MLOOP in RAM ;H=PCHL,L=POP H ;POP H, PCHL at STACK-2, STACK-1 ;addr of MLOOP in HL and stack RAM ;point SP to MLOOP address ; in stack memory ;get next EPROM byte ;store it in RAM</pre>

CDB	L.P	RN
-----	-----	----

	; e=1=0			
	;=====================================			
	;=====================================			
	Wait for user to insert a diskette into the drive 0, and then load that drive's head. Do this first so that the disk has plenty of time to settle. Note that a minidisk will always report that it is ready. Minidisks will hang (later on) waiting for sector OF, until a few seconds after the user inserts a disk.			
	, ; On En ; 1 =			
4C18 AF 4C19 D308 4C1B DB08 4C1D E608 4C1F C2184C	; WAITEN:	xra out in ani jnz	a DENABL DSTAT DRVRDY WAITEN	;boot from disk 0 ;enable disk 0 ;Read drive status ;Diskette in drive? ;no: wait for drive ready
4C22 3E04 4C24 D309		mvi out	a,HDLOAD DCTRL	;Load 8" disk head, or enable ;minidisk for 6.4 Sec
	; in once, then step out until track 0 is detected ; On Exit: b=0			
4C26 018206 4C29 3E01	;	lxi mvi		;20 mS delay 1st time thru ;step in once first
4C2B D309	SKTRK0:	out	DCTRL	;issue step command
	; minim ; the 8	um 43 ms " spec f	step wait inste	at least 20ms to force a ad of 10ms. This meets direction. The minidisk
4C2D 0B 4C2E 78 4C2F B1 4C30 C22D4C	DELAY:	dcx mov ora jnz	b a,b c DELAY	;(5) ;(5) ;(4) ;(10)12 uS/pass
4C33 0C		inr	с	;from now on, the above loop ;goes 1 time.
4C34 DB08 4C36 OF 4C37 OF 4C38 DA344C	WSTEP:	in rrc rrc jc	DSTAT WSTEP	;wait for step to complete ;put MVHEAD bit in carry ;is the servo stable? ;no: wait for servo to settle
4C3B E610 4C3D 3E02 4C3F C22B4C		ani mvi jnz	TRACKO/4 a,STPOUT SKTRKO	;Are we at track 00? ;STEP-OUT command ;no: step out another track

;Exit with b=0

CDBL.PRN Determine if this is an 8" disk or a minidisk, and set c to the correct sectors/track for the detected disk. An 8" disk has 20h sectors, numbered 0-1Fh. A minidisk has 10h sectors, numbered 0-0Fh. ; wait for the highest minidisk sector, sector number OFh 4C42 DB09 CKDSK1: in DSECTR ;Read the sector position SECMSK+SVALID 4C44 E63F ani ;Mask sector bits, and hunt 4C46 FE1E (MDSPT-1)*2 ;..for minidisk last sector cpi 4C48 C2424C jnz CKDSK1 :..only while SVALID is 0 ; wait for this sector to pass ;Read the sector position ;wait for invalid sector 4C4B DB09 CKDSK2: in DSECTR 4C4D 0F rrc 4C4E D24B4C CKDSK2 jnc ; wait for and get the next sector number 4C51 DB09 CKDSK3: in DSECTR ;Read the sector position 4C53 0F ;put SVALID in carry rrc 4C54 DA514C CKDSK3 wait for sector to be valid jc The next sector after sector OFh will be 0 for a minidisk, and 10h for an 8" disk. Adding MDSPT (10h) to that value ; will compute c=10h (for minidisks) or c=20h (for 8" disks). 4C57 E61F 4C59 C610 SECMSK/2 ;Mask sector bits ani MDSPT ; compute SPT adi 4C5B 4F mov c,a ;...and save SPT in c Initialize the ACIA (2SIO port O/Turnkey/UIO). Do this late in the initialization, so that e.g. the 'B' character from UBMON won't get eaten by resetting the ACIA. 4C5C 3E03 ;reset first mvi a, ACRST 4C5E D310 out ACCTRL :then initialize 4C60 3E11 mvi a,ACINIT 4C62 D310 out ACCTRL Initialize the 4PIO _____ 4C64 AF xra а 4C65 D322 out P4CB0 ;Port 0 section B is output 4C67 2F ;All output bits high cma 4C68 D323 P4DB0 out 4C6A 3E2C mvi a, P4CINI ;set up handshake bits 4C6C D322 P4CB0 out Set up to load On Entry: b = 0 (initial sector number) c = SPT (for either minidisk or 8" disk) 1 = 0 (part of DMA address) _____

Page 6

4C6E 65	mo∨	CDBL.PRN h,l	;initial DMA address=0000
	; checksum is ; b = curren ; c = sector	sector over and right, or there l t sector number s/track for this t DMA address	over, until either the have been too many retries kind of disk
4C6F 3E10	; NXTSEC: mvi	a,RETRYS	;(7)Initialize sector retries
	; b = Curren : c = Sector	Read ing retries for t t sector number s/track for this t DMA address	
4C71 317B4D 4C74 F5	RDSECT: lxi push	sp,STACK psw	;(10)(re)initialize the stack ;(11)Remaining retry count
	, Sector Read:	will become ava	r sector specified in b. Data iable 250 uS after -SVALID ID is low for 30 uS.
4c75 db09	FNDSEC: in	DSECTR	;(10)Read the sector position
4C77 E63F	ani		;(7)yes: Mask sector bits
4C79 OF 4C7A B8	rrc cmp	b	;along with -SVALID bit ;(4)sector bits to bits <4:0> ;(4)Found the desired sector ;with -SVALID low? :(10)no: wait for it
4с7в с2754с	jnz	FNDSEC	;(10)no: wait for it
	; ; Test for DMA ; or the next ; Do this here	address that wou page (which conta , while we have s	uld overwrite this RAM code ains the sector buffer stack) some time.
4C7E 117B4D	, lxi	d,SECBUF	;(10)Sector buffer address
4C81 7C 4C82 AA 4C83 E6FE 4C85 3E4F 4C87 CAE14C	mov xra ani mvi jz	a,h d OFEh a,OERMSG RPTERR	;(5)high byte of DMA address ;(4)high byte of RAM code addr ;(7)ignore lsb ;(7)overlay error message ;(10)report overlay error
	, Set up for t , Do this here	he upcoming data , while we have s	move some time.
4C8A E5 4C8B C5 4C8C 018000	,push push lxi	h b b,BPS	;(11)Current DMA address ;(11)Current sector & SPT ;(10)b= init checksum, ;c= byte count for MOVLUP
	Sector Read:	SECBUF is posit overflows at the becomes availab	ctor data into SECBUF at de. ioned in memory such that e e end of the buffer. Read data le 250 uS after -SVALID becomes pop must be << 32 uS per pass.

4C8F DB08 4C91 07 ;(10)Read the drive status DATLUP: in DSTAT (4) New Read Data Available? rlc 4C92 DA8F4C DATLUP ;(10)no: wait for data jc 4C95 DB0A in DDATA ;(10)Read data byte 4C97 12 (7) Store it in sector buffer stax d 4C98 1C (5) Move to next buffer address inr е ;...and test for end ;(10)Loop if more data 4C99 C28F4C jnz DATLUP _____ Sector Read: Step 3. Move sector data from SECBUF into memory at hl. Compute checksum as we go. 8327 cycles for this section e,SDATA and OFFh ;(7)de= address of sector data 4C9C 1E7E mvi ;...within the sector buffer 4C9E 1A 4C9F 77 MOVLUP: ldax d ;(7)Get sector buffer byte ;(7)Store it at the destination mov m,a ;(7)Did it store correctly? ;(10)no: abort w/ memory error 4CA0 BE cmp m 4CA1 C2DF4C MEMERR jnz 4CA4 80 add ;(4)update checksum h 4CA5 47 (5) Save the updated checksum mov b,a 4CA6 13 4CA7 23 4CA8 0D d ;(5)Bump sector buffer pointer inx ;(5)Bump DMA pointer ;(5)More data bytes to copy? inx h dcr С 4CA9 C29E4C ;(10)yes: loop MOVLUP jnz Sector Read: Step 4. Check Marker byte and compare computed checksum against sector's checksum. Retry/abort if wrong Marker byte or checksum mismatch. a=computed checksum 98 cycles for for this section ------;(4)hl=1st trailer byte address 4CAC EB xchq ;de=DMA address 4CAD 4E mov ;(7)get marker, should be FFh c,m (5)c should be 0 now 4CAE 0C inr С 4CAF 23 inx ;(5)(h1)=checksum byte h 4CB0 AE 4CB1 B1 ;(7)compare to computed cksum ;(4)..and test marker=ff xra m ora С 4CB2 C1 h ;(10)Current sector & SPT pop 4CB3 C2D24C ;(10)NZ: checksum error jnz BADSEC Compare next DMA address to the file byte count that came ; from the sector header. Done of DMA address is greater. ;(16)hl gets file size ;(4)put DMA address back in hl 4CB6 2A7C4D lhld SFSIZE 4CB9 EB xchg ...and file size into de ;(4)16-bit subtraction 4CBA 7D mov a,1 4CBB 93 ;(4) sub e 4CBC 7C ;(4)..throw away the result mov a,h

CDBL.PRN

4cbd 9a	sbb	CDBL.PRN d	;(4)but keep carry (borrow)
4CBE D2E34C	jnc	LDDONE	;(10)done loading if hl >= de ;carry will be clear at LDDONE
	Next Sector:	the even sectors	interleaved by two. Read all first, then the odd sectors. will repair the stack.
	, 44 cycles for	the next even o	or next odd sector
4CC1 116F4C 4CC4 D5	,lxi push	d,NXTSEC d	;(10)for compact jumps ;(10)
4CC5 04 4CC6 04	inr inr	b b	;(5)sector = sector + 2 ;(5)
4сс7 78 4сс8 в9 4сс9 D8	mo∨ cmp rc	a,b c	;(5)even or odd sectors done? ;(4)c=SPT ;(5/11)no: go read next sector ;at NXTSEC
	; one 8" sector	to-sector = 28+8 time = 5208 us, sector, no prob	327+98+44=8497 cycles=4248.5 us so with 2:1 interleave, we will lem.
4CCA 0601 4CCC C8	mvi rz	b,01н	;1st odd sector number ;Z: must read odd sectors now ;at NXTSEC
	; C ; S ; C ; C ; C ; C ; C ; C ; C ; C ; C ; C	since we just rea bon't need to wai either, because w revolution going sector 0. (One re step takes a a ma	again. The head to be ready (-MVHEAD), and the entire previous track. It for this step-in to complete we will definitely blow a from the track's last sector to evolution takes 167 mS, and one aximum of 40 uS.) will repair the stack.
4CCD 78 4CCE D309	, mov out	a,b DCTRL	;STEPIN happens to be 01h
4CD0 05 4CD1 C9	dcr ret	b	;start with b=0 for sector 0 ;go to NXTSEC
	; Checksum erro ; already. Othe ; On Entry: ; Top of stac	or: attempt retry erwise, abort, re ck = adress for f	if not too many retries porting the error irst byte of the failing sector
4CD2 3E04 4CD4 D309	; Next on sta ;************************************	a,HDLOAD DCTRL	;Restart Minidisk 6.4 uS timer
4CD4 D309 4CD6 E1 4CD7 F1 4CD8 3D 4CD9 C2714C	pop pop dcr jnz	h psw a RDSECT	;Restore DMA address ;Get retry count ;Any more retries left? ;yes: try reading it again

CDBL.PRN Irrecoverable error in one sector: too many retries. These errors may be either incorrect marker bytes, wrong checksums, or a combination of both. On Entry: hl=RAM adress for first byte of the failing sector _____ ;Checksum error message 4CDC 3E43 mvi a,CERMSG 4CDE 11 'lxi d' opcode to skip db 11H ; .MEMERR and go to RPTERR ;***Error Routine************************ Memory error: memory readback failed On Entry: hl = offending RAM address *********** 4CDF 3E4D MEMERR: mvi a, MERMSG ;Memory Error message ; Fall into RPTERR Entry at RPTERR: Report an error: turn the disk controller off, turn the INTE light on, record the error in RAM at 0000h-0002h, and then loop forever writing the error code (in register a) to all known Terminal ports. On Entry: a = error code hl = offending RAM address Entry at LDDONE: Normal exit: Disable the disk controller and go execute the loaded code at DMAADR. On Entry: Carry bit is cleared ******* 4CE1 47 RPTERR: mov b,a ;error code 4CE2 37 ;remember we had an error stc 4CE3 3E80 :Disable the disk controller LDDONE: mvi a, DDISBL 4CE5 D308 out DENABL 4CE7 D20000 jnc DMAADR ;normal exit: go execute the ;..loaded program 4CEA FB ;Signal error on the INTE LED ei 4CEB 220100 4CEE 78 ;Store the bad address sh1d 1 mov a,b ;recover the error code 4CEF 320000 ;Store the error code sta 0 4CF2 D301 ;SIO ERHANG: out SIODAT 4CF4 D311 ACDATA ;2SIO port 0/Turnkey/UIO out 4CF6 D305 out PIODAT ;PIO 4CF8 D323 P4DB0 out :4PIO ;Keep printing error code 4CFA C3F24C ERHANG jmp 4CFD end